

Opgavebesvarelse vedr. overlevelsesanalyse

Opgaven går ud på at vurdere effekten af azathioprine på overlevelsen hos patienter med primær biliær cirrhose (PBC), ref. Christensen et al. (1985).

Data stammer fra en "clinical trial", hvor 216 patienter er randomiseret til behandling med azathioprine eller placebo, hvorefter de følges over tid indtil død eller censurering. (**OBS:** Der er 9 patienter, der censureres lige fra start, og de er nu pillet ud af datasættet, men de er med i denne opgavebesvarelse).

Filen `pbcc.txt` indeholder følgende variable:

dead: Indikator for død (0: Censurering, 1: Død)

months: Antal måneder fra randomisering til udgang (død eller censurering)

azathiop: Behandling (0: Placebo, 1: Azathioprine)

bilirubin: Serum bilirubin måling ved baseline

Først må vi indlæse data, og samtidig tilføjer vi nogle labels, som gør vores udskrifter mere umiddelbart forståelige:


```
data pbcc;  
infile "http://staff.pubhealth.ku.dk/~lts/basal/data/pbcc.txt"  
    URL firstobs=2;  
INPUT Dead Months Azathiop Bilirubin;  
  
LABEL  
    Dead = "Død (0=censur, 1=død)"  
    Months = "Måneder fra randomisering"  
    Azathiop = "Azathioprine (0~Placebo, 1~Azathioprine)"  
    Bilirubin = "Baseline serum bilirubin"  
;  
RUN;
```

1. *Da der er tale om et randomiseret studie, vurderes behandlingseffekten først uden inddragelse af diverse kovariater.
Vurder effekten af azathioprine ud fra*

(a) *Kaplan-Meier kurver*

Kaplan-Meier kurver er nonparametriske estimater for overlevelseskurverne (nonparametrisk = ingen antagelser om fordelingen af dødstidspunkterne).

```
proc phreg plots(overlay=row cl)=survival data=pbcr;  
  model months*dead(0)= ;  
  strata azathiop;  
run;
```



De er her tegnet med punktvisse konfidensgrænser, fordi det oftest er nødvendigt at have konfidensgrænserne med for at kunne vurdere, om der er klar forskel på grupperne....

Vi ser, at der er adskillige overkrydsninger mellem de to overlevelseskurver og ingen iøjnefaldende forskelle undtagen, at de måske divergerer lidt mod slut.

(b) *et passende test*

Vi skal se på tid til død, under hensyntagen til, at der forekommer censureringer ($dead=0$), og vi skal sammenligne to grupper. Det relevante test er et *log-Rank test*:

```

proc phreg data=pb;
  class azathiop(ref="0") / param=glm;
  model months*dead(0)=azathiop / ties=discrete rl;
estimate "placebo vs azathiop" azathiop -1 1 / exp cl;
estimate "azathiop vs placebo" azathiop 1 -1 / exp cl;
run;

```

Vi finder nedenstående output, der ikke viser nogen signifikant forskel på de to grupper ($P=0.4936$ for log-rank test, fundet under Testing Global Null Hypothesis: $BETA=0$ ud for Score).

The PHREG Procedure

Model Information

Data Set	WORK.PBC	
Dependent Variable	months	Måneder fra randomisering
Censoring Variable	dead	Død (0=censur, 1=død)
Censoring Value(s)	0	
Ties Handling	DISCRETE	
Number of Observations Read	216	
Number of Observations Used	216	

Summary of the Number of Event and Censored Values

Total	Event	Censored	Percent Censored
216	103	113	52.31

Testing Global Null Hypothesis: $BETA=0$

Test	Chi-Square	DF	Pr > ChiSq
Likelihood Ratio	0.4671	1	0.4943
Score	0.4687	1	0.4936
Wald	0.4680	1	0.4939

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
azathiop 1	1	-0.13578	0.19849	0.4680	0.4939
azathiop 0	0	0	.	.	.

Parameter	Hazard Ratio	95% Hazard Ratio Confidence Limits
azathiop 1	0.873	0.592 1.288
azathiop 0	.	.

Analysis of Maximum Likelihood Estimates

Parameter	Label
azathiop 1	Azathioprine (0~Placebo, 1~Azathioprine) 1
azathiop 0	Azathioprine (0~Placebo, 1~Azathioprine) 0

Estimate

Label	Estimate	Standard		z Value	Pr > z	Alpha
		Error				
placebo vs azathiop	0.1358	0.1985		0.68	0.4939	0.05
azathiop vs placebo	-0.1358	0.1985		-0.68	0.4939	0.05

Label	Lower	Upper	Exponentiated
placebo vs azathiop	-0.2532	0.5248	1.1454
azathiop vs placebo	-0.5248	0.2532	0.8730

Label	Exponentiated	
	Lower	Upper
placebo vs azathiop	0.7763	1.6901
azathiop vs placebo	0.5917	1.2882

- (c) *et estimat for hazard ratioen for azathiop-behandlede vs. placebo-behandlede, med konfidensinterval*

Da vi havde valgt placebo som reference (`azathiop=0`), fik vi hazard ratio for de azathioprine-behandlede vs de placebo-behandlede direkte skrevet ud i outputtet ovenfor som 0.873, med konfidensgrænser (0.592, 1.288).

Hvis vi i stedet vil beskrive den øgede hazard for de placebo-behandlede, skal vi udregne $\exp(0.13578) = 1.145$, og tilsvarende med konfidensgrænserne. I stedet har vi tilføjet en `estimate`-sætning til at foretage denne udregning for os, og vi ser, at konfidensgrænserne bliver (0.776, 1.690).

I begge tilfælde er der tale om konfidensgrænser, der indeholder 1, svarende til, at der *ikke* er signifikant forskel på dødeligheden i de to grupper.


2. *Serum bilirubin niveauet er generelt en meget stærk prognostisk faktor for patienter med leversygdom. Det er derfor relevant at undersøge betydningen af baseline serum bilirubin for prognosen hos patienter med primær biliær cirrhose.*

Vurder først ud fra nedenstående kriterier, om serum bilirubin bør transformeres:

- (a) *ved at se på fordelingen af serum bilirubin – er der ekstreme observationer?*

Et histogram over alle værdierne set for bilirubin viser tydeligt, at der er tale om en meget skæv fordeling:


```
proc sgplot data=pbcr;  
  histogram bilirubin;  
run;
```



Denne skævhed er ikke noget problem i sig selv, da der ikke er noget som helst krav til fordelingen af en kovariat. **Men:** I en så skæv fordeling kan der være meget indflydelsesrige observationer blandt de få, meget høje værdier, og det kan derfor være meget ustabil at bibeholde denne skala. Logaritmetransformation kan bruges til at “trække” ekstremt høje værdier ind mod midten.

Vi ser derfor også på en logaritmetransformeret bilirubin. Her er valgt logaritmen med grundtal 2: $\log_2\text{bili} = \text{LOG2}(\text{bilirubin})$;

```
proc sgplot data=pbcr;  
  histogram log2bili;  
run;
```



Her ser vi en meget mere jævn fordeling uden meget ekstreme værdier. Det er derfor relevant at undersøge, om utransformeret bilirubin eller logaritmetransformeret bilirubin er mest relevant som prognostisk variabel – for det vigtigste er jo, at det er en passende model for sammenhængen mellem bilirubin og dødsraten.

- (b) *ved at inkludere serum bilirubin både utransformeret og passende transformeret på samme tid*

Dette kan gøres enten med eller uden behandlingen som kovariat, og vi ser på begge muligheder nedenfor. Først uden behandlingseffekt:

```
proc phreg data=pbcr;
  model months*dead(0)=bilirubin log2bili;
run;
```

der giver os outputtet

The PHREG Procedure

Model Information

Data Set	WORK.PBC	
Dependent Variable	months	Måneder fra randomisering
Censoring Variable	dead	Død (0=censur, 1=død)

Censoring Value(s) 0
Ties Handling BRESLOW

Number of Observations Read 216
Number of Observations Used 216

Summary of the Number of Event and Censored Values

Total	Event	Censored	Percent Censored
216	103	113	52.31

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
bilirubin	1	-0.00184	0.00195	0.8910	0.3452
log2bili	1	0.80690	0.13978	33.3248	<.0001

Parameter	Hazard Ratio	Label
bilirubin	0.998	Baseline serum bilirubin
log2bili	2.241	

Ovenstående output viser klart, at

- den logaritmetransformerede version af bilirubin er den med stærkest statistisk signifikans, mens den utransformerede kan udelades
- sammenhængen flader en smule *mere* ud end logaritme-transformationen (koefficienten til bilirubin er negativ)

Og så til analysen *med* behandlingseffekt:

```
proc phreg data=pbcr;
  class azathiop(ref="0") / param=glm;
  model months*dead(0)=azathiop bilirubin log2bili;
run;
```

der giver outputtet

The PHREG Procedure

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
azathiop 1	1	-0.50982	0.20592	6.1297	0.0133
azathiop 0	0	0	.	.	.

bilirubin	1	-0.00162	0.00196	0.6778	0.4104
log2bili	1	0.84117	0.14212	35.0331	<.0001

Parameter		Hazard Ratio	Label
azathiop	1	0.601	Azathioprine (0~Placebo, 1~Azathioprine) 1
azathiop	0	.	Azathioprine (0~Placebo, 1~Azathioprine) 0
bilirubin		0.998	Baseline serum bilirubin
log2bili		2.319	

der ses at vise stort set det samme, for så vidt angår effekten af bilirubin.

Vi noterer os dog, at behandlingseffekten nu er blevet signifikant - hvilket vi skal komme tilbage til nedenfor.

- (c) *ved at modellere effekten af serum bilirubin ved hjælp af en lineær spline*

Dette er svært... Først skal vi have fundet nogle fornuftige steder til opdeling af splinen, altså nogle steder (værdier af bilirubin), hvor vi tillader kurven at knække. I mangel af bedre viden om forløbet, kan man argumentere for at opdele efter kvartiler af bilirubin, men for at skaffe ca. lige mange dødsfald i hver gruppe, vil vi benytte kvartilerne udregnet udelukkende blandt de døde.

Disse findes først ved at skrive:

```
proc means N Q1 median Q3 data=pbcr; where dead=1;
var bilirubin;
run;
```

hvorved vi får

Analysis Variable : bilirubin Baseline serum bilirubin

N	Lower Quartile	Median	Upper Quartile
103	30.0000000	54.0000000	125.0000000

Vi vælger altså tærskelværdierne 30, 54 og 125 for bilirubin, og oppe øverst i programmet (inden første `run;`) skriver vi de 3 definitioner ind:

```
bili25=(bilirubin-30)*(bilirubin>30);
bili50=(bilirubin-54)*(bilirubin>54);
bili75=(bilirubin-125)*(bilirubin>125);
```

hvorefter vi er klar til at benytte disse ekstra 3 variable sammen med selve `bilirubin`. Vi tilføjer en `TEST`-sætning, der undersøger, om de 3 ekstra variable kan undværes på en gang, idet dette er et test for linearitet af selve `bilirubin`:

```
proc phreg data=pbcr;
  class azathiop(ref="0") / param=glm;
  model months*dead(0)=azathiop bilirubin bili25 bili50 bili75;
  output out=tegn2 xbeta=spline;
  Testlinearitet: TEST bili25=bili50=bili75=0;
run;
```

Herved får vi outputtet:

The PHREG Procedure

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
azathiop 1	1	-0.50759	0.20678	6.0261	0.0141
azathiop 0	0	0	.	.	.
bilirubin 1	1	0.07991	0.02225	12.9044	0.0003
bili25	1	-0.06220	0.03429	3.2907	0.0697
bili50	1	-0.00361	0.02090	0.0299	0.8628
bili75	1	-0.01092	0.00657	2.7665	0.0963

Analysis of Maximum Likelihood Estimates

Parameter	Hazard Ratio	Label
azathiop 1	0.602	Azathioprine (0~Placebo, 1~Azathioprine) 1
azathiop 0	.	Azathioprine (0~Placebo, 1~Azathioprine) 0
bilirubin	1.083	Baseline serum bilirubin
bili25	0.940	
bili50	0.996	
bili75	0.989	

Linear Hypotheses Testing Results

	Wald		
Label	Chi-Square	DF	Pr > ChiSq
Testlinearitet	34.3224	3	<.0001

Output fra PHREG viser, at der ikke er en lineær sammenhæng med baseline serum bilirubin! Testet for linearitet (**Testlinearitet**) giver $P < .0001$.


Det numeriske test bør dog aldrig stå alene – man skal altid tegne den lineære spline. Hvis P -værdien er lav, men det er sammen med en meget savtakket kurve omkring en generelt lineær trend, så kan en lineær sammenhæng stadig være den mest relevante model at bruge.

Vi har i koden ovenfor dannet et nyt datasæt, kaldet **tegn**, som indeholder de predikterede værdier på log hazard skala, kaldet **spline** (**xbeta=spline**). Vi kan tegne disse op som funktion af bilirubin og behandling ved at skrive:

```
proc sort data=tegn;
  by azathiop bilirubin;
run;

proc sgplot data=tegn;
  series Y=spline X=bilirubin / group=azathiop markers;
run;
```

hvorved vi får figuren:



Vi ser her en påfaldende lighed med en logaritmefunktion. Hvis man kørte hele proceduren med en lineær spline igennem med `log2bili` som variabel i stedet for `bilirubin`, ville man faktisk finde en næsten perfekt lineær sammenhæng.

3. *Estimer og fortolk baseline serum bilirubins betydning for dødsraten.*

Alt under spørgsmål 2 viser, at vi bør se på logaritmetransformeret bilirubin. Vi vælger $\log_2$ -transformeret bilirubin, fordi så får vi i outputtet vist effekten af en fordobling af serum bilirubin direkte som effektestimatet.

```
proc phreg data=pbcr;
  model months*dead(0)=azathiop log2bili / rl;
run;
```

Vi finder så outputtet:

The PHREG Procedure

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
-----------	----	--------------------	----------------	------------	------------

azathiop	1	1	-0.52028	0.20607	6.3743	0.0116
azathiop	0	0	0	.	.	.
log2bili		1	0.74246	0.07731	92.2356	<.0001

Parameter		Hazard Ratio	95% Hazard Ratio Confidence Limits
azathiop	1	0.594	0.397 0.890
azathiop	0	.	.
log2bili		2.101	1.806 2.445

Analysis of Maximum Likelihood Estimates

Parameter	Label
azathiop 1	Azathioprine (0~Placebo, 1~Azathioprine) 1
azathiop 0	Azathioprine (0~Placebo, 1~Azathioprine) 0

En dobbelt så høj baseline serum bilirubin giver en 2.10 (95% kofidensinterval: 1.81 til 2.45) gange større dødsrate, altså ca. en fordobling af dødsraten.

4. *Estimer og fortolk effekten af azathioprine, justeret for effekten af bilirubin.*

Vi ser nu på det andet aspekt af analysen ovenfor, nemlig effekten af behandling med azathioprin. Vi konstaterede allerede tidligere, at denne var blevet signifikant efter justering med bilirubin, og P-værdien i analysen ovenfor er 0.012. Vi skal nu kvantificere effekten.

Konklusionen ses at være, at den estimerede effekt af behandling med azathioprine er en dødsrate på kun 59.4% (95% konfidensinterval fra 39.7% til 89.0%) af dødsraten blandt placebo-behandlede med samme baseline serum bilirubin.

Det var lidt af en forskel fra den ujusterede analyse...!

5. *Hvordan ville du afrapportere resultaterne i en artikel?*

Her bør man tage i betragtning,

- (a) at Cochrane centeret har vist, at justerede analyser i gennemsnit overestimerer de faktiske behandlingseffekter (sikkert fordi folk

tror på behandlingseffekten og derfor tror, at nu har de fundet den “rigtige” model, når den estimerede effekt tilfældigvis bliver stærkere).

- (b) at randomisering ikke sikrer, at man får helt ens fordelinger af confoundere, kun at der ikke er nogen systematiske forskelle til fordel for en bestemt behandlingsgruppe
- (c) hvad man selv skrev i protokollen (så man undgår fisketure)

6. *Lav modelkontrol for Cox modellen, primært for effekten af den randomiserede behandling, men hvis der er tid, også for effekten af bilirubin. Modelkontrollen indbefatter:*

- (a) *Grafisk modelkontrol af proportionale rater for behandling*


Vi skal her se på antagelsen om proportionale intensiteter, hvilket gøres ved at tegne “log minus log” af overlevelseskurverne. Hvis der er proportionale intensiteter, skal disse kurver være parallelle.

Vi ser først på behandlingseffekten, som er den nemmeste, fordi der kun er to niveauer (azathioprin eller placebo). Vi benytter den automatiske tegning i SAS af de kumulerede rater ved at skrive det kryptiske `PLOTS(OVERLAY=ROW CL)=CUMHAZ` i `PHREG`-kaldet. Desværre giver dette dog ikke de kumulerede rater på log-skala, og der skal derfor her vurderes proportionalitet, og ikke parallellitet.

Derfor gemmes også logaritmen af de kumulerede rater i det nye datasæt `ud1` under navnet `logCumrate`, som vi så siden kan benytte til egne figurer. Totalt set skriver vi:

```
proc phreg plots(overlay=row cl)=cumhaz data=pb;
  model months*dead(0)=log2bili;
  strata azathiop;
  baseline out=ud1 loglogs=logCumrate;
run;
```

og får den automatiske figur




Det er svært ud fra denne at vurdere proportionalitet, men det ser ud som om kurverne er mere eller mindre ens i starten, for først at divergere omkring 60 måneder. Dette tyder ikke på god proportionalitet.

I stedet tegner vi så de kumulerede rater på logaritmisk skala ved at benytte koden:

```
proc sgplot data=ud1;
series Y=logCumrate X=months / group=azathiop;
run;
```

hvorved vi får figuren



De to kurver ses at krydse over hinanden flere gange i det første $1\frac{1}{2}$ år. I perioden fra $1\frac{1}{2}$ til 5 år er der lidt variende *lodret* afstand mellem kurverne, mens kurverne og bliver mere og mere forskellige efter 5 år.

Ofte kan det være næsten umuligt at bedømme den *lodrette* afstand mellem kurverne i den først del af perioden, fordi raterne ofte er særligt høje i begyndelsen, så kurverne er meget stejle. I så fald er man nødt til også at tegne med eksempelvis en logaritmisk tidsakse. Det er ikke nødvendigt her, fordi kurverne krydser frem og tilbage over hinanden, og overkrydsningerne kan stadig ses, selvom kurverne er relativt stejle.

(b) *Grafisk modelkontrol af proportionale rater bilirubin*

For at checke antagelsen om proportionale intensiteter for bilirubin, er vi nødt til at foretage en inddeling i et par intervaller (vi kan jo ikke checke proportionalitet af kurver svarende til hver enkelt forekommende værdi af bilirubin). Vi vælger her at opdele i 3 intervaller, svarende til fraktilerne 33.33% og 66.67%, igen kun for de døde. Disse findes således:

```
proc univariate data=pbcr; where dead=1;
    var bilirubin;
    output out=regn pctlpre=P_ pctlpts=33.33, 66.67;
run;
```

```
proc print data=regn;
run;
```


Obs	P_33_33	P_66_67
1	37	92


Vi opdeler altså efter bilirubin-værdierne 37 og 92, og det kan gøres direkte i PHREG-kaldet, hvor vi med det samme tilføjer plottet af de kumulerede rater på logaritmisk skala, som forklaret ovenfor:

```
proc phreg plots(overlay=row cl)=cumhaz data=pb;
  model months*dead(0)=azathiop;
  strata bilirubin(37,92);
  baseline out=ud2 loglogs=logCumrate;
run;
```

```
proc sgplot data=ud2;
  series Y=logCumrate X=months / group=bilirubin;
run;
```

Herved får vi to figurer





Bortset fra det første års tid synes proportionaliteten at være vældig god for bilirubin.

De næste 3 punkter tager vi under et:

- (c) *Numerisk test af proportionale rater for behandling*
- (d) *Numerisk test af proportionale rater for bilirubin*
- (e) *Check af linearitetsantagelsen for bilirubin (på den valgte skala)*

Vi kan udnytte automatiske tests i SAS ved at skrive en ekstra sætning, kaldet **assess**. Her betyder keyword PH, at der ønskes test for proportional hazards, mens VAR=(log2bili) skrives for at få en vurdering af lineariteten for log2bili. Den totale kode bliver

```
proc phreg data=pbcr;
  class azathiop(ref="0") / param=glm;
  model months*dead(0)=azathiop log2bili / rl;
  assess var=(log2bili) ph / resample seed=106165;
run;
```

Denne kode frembringer det ekstra output:


Supremum Test for Functional Form				
Variable	Maximum Absolute Value	Replications	Seed	Pr > MaxAbsVal
log2bili	4.5171	1000	106165	0.8180


Supremum Test for Proportionals Hazards Assumption				
Variable	Maximum Absolute Value	Replications	Seed	Pr > MaxAbsVal
azathiop1	0.9640	1000	106165	0.2910
log2bili	0.9151	1000	106165	0.2400

Her ses ingen tegn på problemer med hverken antagelsen om proportionale rater ($P=0.29$ hhv $P=0.24$ for azathiop hhv bilirubin), eller med linearitetsantagelsen for bilirubin på log-skala ($P=0.82$).

Man får i tilgift figurer, som kan give fingerpeg om, hvordan afvigelserne fra antagelserne evt kunne se ud.

Først for antagelsen om proportionale rater:






Den øverset figur, for behandlingseffekten, viser, at personer med “høje værdier” af `azathiop1` (dvs. de behandlede) dør lidt for hurtigt til at starte med (det var jo også her, de kumulerede rater var nærmest ens). Det kan jo være, at behandlingseffekten først sætter ind efter et stykke tid....

For bilirubin ser vi ligeledes, at dem med høj værdi af bilirubin dør “for hurtigt” til at starte med, hvilket kunne være en effekt af, at det drejer sig om en værdi målt ved baseline, og denne har formentlig større prognostisk værdi i starten (da den kan ændre sig hen ad vejen).

Vi ser dernæst på figuren svarende til linearitetsantagelsen for `log2bili`:



Denne figur viser kun ukarakteristiske svingninger, svarende til en rigtig god linearitetsantagelse, som vi også fandt ud af i spørgsmål 3c, hvor vi modellerede effekten som en lineær spline.

Ovenstående er en nem metode til at lave modelkontrol, men man kan supplere med mere specifikke tests, i form af tidsvarierende effekter.

Man kan så vælge nogle passende tidspunkter, hvor proportionalitetsfaktoren kan få lov til at ændre sig. En god måde at vælge disse opdelingstidspunkter er at undersøge, om de kumulerede rater er nogenlunde lineære i nogle intervaller. Hvis de kumulerede rater er nogenlunde lineære, så er raterne nogenlunde konstante i de samme intervaller. Hvis raterne er konstante i begge grupper i et interval, er de selvfølgelig også proportionale i hele det interval.

Hvis vi f.eks. ikke antager proportionale intensiteter for behandling i *hele* tidsrummet, men tillader proportionalitetskonstanten at variere mellem f.eks. 3 tidsintervaller (her valgt med tærskelværdier 12 og 56 måneder, baseret på figuren fra spørgsmål 6a), kan vi skrive:

```
proc phreg data=pbcr;
  model months*dead(0)=log2bili az1 az2 az3 / rl;
```

```

if months<=12 then az1=azathiop; else az1=0;
if 12<months<=56 then az2=azathiop; else az2=0;
if 56<months then az3=azathiop; else az3=0;
NumCheckAza: test az1=az2=az3;
run;

```

og vi finder herved:

Analysis of Maximum Likelihood Estimates					
Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
log2bili	1	0.74400	0.07757	91.9999	<.0001
az1	1	-0.05003	0.43591	0.0132	0.9086
az2	1	-0.42939	0.27472	2.4430	0.1180
az3	1	-1.15512	0.41058	7.9151	0.0049

Linear Hypotheses Testing Results				
Label	Chi-Square	DF	Pr > ChiSq	Wald
NumCheckAza	3.7223	2	0.1555	

Her finder vi altså heller ikke nogen indikation af, at proportion-
alitetskonstanterne skulle afvige væsentligt fra hinanden.

I stedet for at basere opdelingen på det visuelle indtryk af stykke-
vis linearitet for den kumulerede rate, kan man eksempelvis vælge
at opdele tidsintervallerne, så der bliver ca. lige mange døde i hvert
tidsinterval, fordi så er der ca. lige stor statistisk styrke i hvert in-
terval. Dette prøver vi for effekten af bilirubin:

```

proc univariate data=pbcs; where dead=1;
var months;
output out=regn pctlpre=P_ pctlpts=33.33, 66.67;
run;

```

```

proc print data=regn;
run;

```

Obs	P_33_33	P_66_67
1	18.95	50.16

Benytter vi således 19 og 50 måneder som tærskelværdier, skriver vi:

```
proc phreg data=pb;
  model months*dead(0)=azathiop log2bili1 log2bili2 log2bili3 / rl;
  if months<=19 then log2bili1=log2bili; else log2bili1=0;
  if 19<months<=50 then log2bili2=log2bili; else log2bili2=0;
  if 50<months then log2bili3=log2bili; else log2bili3=0;
  NumCheckBili: test log2bili1=log2bili2=log2bili3;
run;
ods graphics off;
```

hvorved vi finder

The PHREG Procedure

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
azathiop	1	-0.50679	0.20618	6.0418	0.0140
log2bili1	1	0.82555	0.11904	48.0981	<.0001
log2bili2	1	0.58916	0.12939	20.7340	<.0001
log2bili3	1	0.81331	0.15332	28.1375	<.0001

Linear Hypotheses Testing Results

Label	Wald Chi-Square	DF	Pr > ChiSq
NumCheckBili	2.1604	2	0.3395

Det ser således ud til, at proportionalitetsantagelsen (=konstant effekt) for serum bilirubin holder rimeligt godt henover hele perioden.

Reference:

Christensen E, Neuberger J, Crowe J, Altman DG, Popper H, Portmann B, et al. Beneficial effect of azathioprine and prediction of prognosis in primary biliary cirrhosis. Final results of an international trial. *Gastroenterology* 1985;89(5):1084-91.